# Maintainer: Andrea Dalle Vacche adallevacche@bloomberg.net

_realname=Math-Int64
pkgname=perl-${_realname}
pkgver=0.54
pkgrel=3
pkgdesc="Math::Int64 - Manipulate 64 bits integers in Perl"
arch=('any')
license=('GPL' 'PerlArtistic')
url="https://metacpan.org/release/Math-Int64"
depends=('perl')
groups=('perl-modules')
options=('!emptydirs')
source=("https://www.cpan.org/authors/id/S/SA/SALVA/${_realname}-${pkgver}.tar.gz")
sha256sums=('dcfc51e698437ea6b9cefe0276215c56cdb6a7f85e3e24a2b6b4189f1960d351')

prepare() {
  cd ${_realname}-${pkgver}
}

build() {
  cd ${_realname}-${pkgver}
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
